



JEUX EN BASIC SUR

TRS-80

COULEUR



JEUX EN BASIC

Auteur : Pierre **MONSAUT**
Illustrations : Jean-François Pénichoux
Photo Lucky Mômes

Tous les efforts ont été faits pour fournir dans ce livre une information complète et exacte. Néanmoins, SYBEX n'assume de responsabilités ni pour son utilisation, ni pour les contrefaçons de brevets ou atteintes aux droits de tierces personnes qui pourraient résulter de cette utilisation.

Copyright version originale © 1984, Sybex

Tous droits réservés. Toute reproduction même partielle, par quelque procédé que ce soit, est interdite sans autorisation préalable. Une copie par xérographie, photographie, film, bande magnétique ou autre, constitue une contrefaçon passible des peines prévues par la loi sur la protection des droits d'auteur.

JEUX EN BASIC

PIERRE MONSAUT



Paris • Berkeley • Düsseldorf





Table des matières

Squash	11
Blitz	15
D.C.A.	19
Trace	23
Slalom	27
Ramasse-miettes	31
Tank	35
Alphabet	39
Atterrissage	43
Chasse au canard	47
Serpent	51
Simon	55
Crabes	59
Grand-prix	65
Casse-briques	69
Poursuite	73
Robots	79
Parachute	85



Avant-Propos

Les jeux présentés dans ce livre utilisent des fonctions propres à votre micro-ordinateur : son, couleurs, graphismes animés, etc. Pour cette raison, ces programmes ne fonctionnent que sur l'ordinateur pour lequel ils ont été conçus et sont difficilement adaptables à des matériels différents.

Au-delà du jeu lui-même, l'étude de ces programmes vous enseignera de nombreuses techniques de programmation qui vous seront très utiles pour développer vos propres programmes.

NOTES CONCERNANT LES PROGRAMMES

Les programmes doivent être tapés tels qu'ils sont reproduits, à l'exception des commentaires (lignes commençant par l'instruction REM) qui ne sont pas nécessaires à leur fonctionnement. Il est cependant conseillé de conserver la ligne contenant le nom des programmes (ligne 10) afin de pouvoir les identifier ultérieurement. Ils doivent être enregistrés sur cassette pour éviter d'avoir à les retaper à chaque utilisation. Du fait de l'utilisation fréquente de l'instruction POKE, il est prudent d'enregistrer les programmes *avant* de les faire tourner. En effet, une simple faute de frappe peut parfois entraîner un blocage de l'ordinateur lors du lancement d'un programme. Dans certains cas, vous serez contraint de débrancher et de rebrancher l'ordinateur pour le remettre en marche et votre programme sera perdu si vous ne l'avez pas enregistré auparavant. Dans le cas contraire, vous pourrez le recharger en mémoire, chercher l'erreur (ou les erreurs), puis sauvegarder la version corrigée. Utilisez pour cela l'instruction

CSAVE "< nom >"

en remplaçant < nom > par le nom du programme à enregistrer.

Exemple : pour enregistrer le programme BLITZ, tapez

CSAVE "BLITZ"

et pressez la touche RETURN après avoir appuyé sur les touches PLAY et RECORD du magnétophone. Pour recharger un programme, rembobinez la cassette jusqu'au début du programme (certains magnétophones possèdent un compteur, très utile si vous avez pris soin de noter le nombre indiqué avant l'enregistrement) et tapez CLOAD "< nom >". Si vous voulez, par exemple, recharger le programme BLITZ, tapez

CLOAD "BLITZ"

et pressez la touche RETURN. Vous pouvez aussi taper simplement


CLOAD (return)

(Pour connaître la différence entre ces deux commandes, reportez-vous à votre manuel d'instructions.)

Afin de réduire au maximum les risques d'erreurs, les listings des programmes ont été reproduits par photographie. Si vos programmes ne fonctionnent pas, vérifiez-les attentivement en considérant plus particulièrement les points suivants :

- lignes oubliées,
- numéros de lignes erronés,
- erreurs de frappe (attention en particulier à ne pas confondre la lettre O et le chiffre 0 ou la lettre I et le chiffre 1).

Tous les jeux présentés fonctionnent en BASIC COULEUR sauf Parachute et Atterrissage qui nécessitent le BASIC COULEUR ETENDU.



SQUASH

Grâce à votre ordinateur, vous pouvez jouer au squash assis dans un fauteuil. La raquette se déplace à l'aide du joystick ou des touches <A>, <S> et barre d'espace. Vous disposez de dix balles que vous devez garder en jeu le plus longtemps possible. Chaque balle renvoyée rapporte un point.

```
10 REM *****
20 REM * SQUASH *
30 REM *****
40 GOTO 280
50 L=INT(JOYSTK(0)/2+444)
60 IF ABS(L-R)>1 THEN R=R+2*SGN(
L-R)
70 RETURN
80 D#=INKEY#
90 D=2*((D#="A")-(D#="S"))
100 IF D<>0 THEN DO=D
110 IF D#=" " THEN DO=0
120 R=R+DO
130 RETURN
140 V=V+DV
144 REM
```

```
145 REM BOUCLE PRINCIPALE
146 REM
150 H=H+DH
160 POKE B+MM,CN
170 B=V*32+H
180 POKE B+MM,CB
190 IF V=13 AND ABS(R-29-B)>1 TH
EN 540
200 IF V=13 THEN S=S+1:SOUND 1,1
210 IF V=1 OR V=13 THEN DV=-DV
220 IF H=1 OR H=30 THEN DH=-DH
230 ON JS GOSUB 50,80
240 IF R<446 THEN R=446
250 IF R>475 THEN R=475
260 PRINT@ R,R#;
270 GOTO 140
274 REM
275 REM INITIALISATION
276 REM
280 CLS
290 GOSUB 480
300 FOR I=1024 TO 1055
310 POKE I,175
320 NEXT I
330 FOR I=1 TO 12
340 POKE I*32+1024,175
350 POKE I*32+1055,175
360 NEXT I
370 R=463
380 R#=CHR$(32)+CHR$(32)+CHR$(13
1)+CHR$(131)+CHR$(131)+CHR$(32)+
CHR$(32)
390 CB=79
400 H=RND(28)+1
410 V=12
420 B=V*32+H
430 CN=96
440 MM=1024
```

```
450 DV=-1
460 DH=(RND(2)-1.5)*2
470 GOTO 140
480 PRINT@ 203,"JOYSTICK ?";
490 D$=INKEY$
500 IF D$="" THEN 490
510 IF D$="0" THEN JS=1 ELSE JS=
2
520 CLS
530 RETURN
534 REM
535 REM PERDU
536 REM
540 NB=NB+1
550 IF NB=11 THEN 640
560 POKE B+MM,CN
570 FOR I=1 TO 3
580 SOUND 1,1
590 FOR J=1 TO 100
600 NEXT J
610 NEXT I
620 DO=0
630 GOTO 400
634 REM
635 REM FIN
636 REM
640 PRINT@ 166,"SCORE :";S;
650 IF S>R1 THEN R1=S
660 PRINT@ 230,"RECORD :";R1;
670 PRINT@ 294,"UNE AUTRE ?";
680 FOR I=1 TO 500
690 NEXT I
700 NB=0
710 S=0
720 D$=INKEY$
730 IF D$="" THEN 720
740 IF D$<>"N" THEN 40
750 END
```




BLITZ

Votre mission est de détruire la ville que vous survolez afin de pouvoir atterrir. A chaque passage, votre avion vole un peu plus bas. Vous ne pouvez larguer une bombe (en appuyant sur une touche quelconque) que lorsque la bombe précédente a atteint son objectif ou le sol. Lorsque votre avion a atterri (ou lorsqu'il s'est écrasé contre un immeuble), le score est affiché ainsi que le record du jour. Si ce jeu vous paraît trop difficile, vous pouvez changer les limites de la ville (6 et 26, ligne 100).

```
10 REM *****
20 REM * BLITZ *
30 REM *****
34 REM
35 REM INITIALISATION
36 REM
39 REM A#=AVION
40 A#=CHR$(128)+CHR$(155)+CHR$(1
47)
49 REM B#=BOMBE
50 B#=CHR$(145)
58 REM H=POSITION DE L'AVION ET
59 REM SCORE
60 H=0
69 REM B=POSITION DE LA BOMBE
70 B=0
80 B1=B
```

```
90 CLS 0
94 REM
95 REM AFFICHAGE DE LA VILLE
96 REM
99 REM 6 ET 26 : LIMITES VILLE
100 FOR I=6 TO 26
110 C=RND(7)+151
114 REM LIGNE 120: REMPLACER B
115 REM PAR VALEUR SUPERIEURE
116 REM POUR DIMINUER LA HAUTEUR
117 REM DES IMMEUBLES
120 FOR J=15 TO RND(4)+8 STEP-1
130 PRINT@ J*32+I,CHR$(C);
140 NEXT J
150 NEXT I
154 REM
155 REM BOUCLE PRINCIPALE
156 REM
159 REM AFFICHAGE AVION
160 PRINT@ H,A$;
169 REM AVION ECRASE?
170 IF PEEK(1027+H)<>128 THEN 24
0
179 REM TIR
180 IF INKEY$<>" " AND B=0 THEN B
=H+33
189 REM BOMBE ATTEINT LE SOL
190 IF B<>0 THEN GOSUB 360
199 REM PAS DE BOMBE LACHEE?
200 IF B=0 THEN GOSUB 400
209 REM AVANCE AVION
210 H=H+1
219 REM AVION POSE?
220 IF H=507 THEN 240
230 GOTO 160
234 REM
235 REM AVION POSE OU ECRASE
236 REM
```

```
239 REM RECORD BATTU?
240 IF H>R THEN R=H
250 PRINT @ 3,"SCORE :";H,
260 PRINT "RECORD :";R;
270 FOR I=1 TO 100
280 NEXT I
290 R#=INKEY#
300 PRINT@ 73,"UNE AUTRE ?";
310 R#=INKEY#
320 IF R#="" THEN 310
330 IF R#<>"N" THEN 40
340 CLS
350 END
354 REM
355 REM BOMBE LACHEE
356 REM
359 REM SOL ATTEINT?
360 IF B>=510 THEN B=0
365 REM AFFICHAGE BOMBE
370 PRINT@ B1,CHR$(128);
380 IF B<>0 THEN PRINT@ B,B#;:B1
=B:B=B+32
390 RETURN
394 REM
395 REM DELAI FOUR RALENTIR
396 REM L'AVION SI AUCUNE BOMBE
397 REM N'EST LACHEE
398 REM
400 FOR I=1 TO 20
410 NEXT I
420 RETURN
```




D.C.A.

Les rôles sont maintenant inversés. Vous manœuvrez la D.C.A. et devez essayer d'abattre les avions qui passent au-dessus de vous. Pour tirer, utilisez n'importe quelle touche. Vous disposez au départ de dix missiles. Si vous abattez huit avions, vous obtenez un bonus de huit points et huit missiles supplémentaires.

```
10 REM *****
20 REM * D.C.A. *
30 REM *****
40 CLEAR 250
50 GOTO 490
60 A#=RIGHT$(A#,1)+LEFT$(A#,31)
70 B#=RIGHT$(B#,31)+LEFT$(B#,1)
80 PRINT@ 0,A#;
90 PRINT@ 64,B#;
100 R#=INKEY#
110 IF R#<>" AND M=495 THEN M=4
63:NM=NM-1
120 IF M<>495 THEN M=M-64:PRINT@
M,M#;:PRINT@ M+64,0#;
130 IF M<>143 THEN 210
140 IF POINT (32,5)<>0 THEN 210
150 PRINT@ 79,CHR$(191);
160 PRINT@ M,0#;
```

```
170 SOUND 1,1
180 S=S+1
190 B$=LEFT$(B$,14)+01$+RIGHT$(B
$,13)
200 GOTO 280
210 IF M<>79 THEN 300
220 IF POINT(30,1)<>0 THEN 300
230 PRINT@ 15,CHR$(191);
240 S=S+1
250 PRINT@ M,0$;
260 SOUND 1,1
270 A$=LEFT$(A$,13)+01$+RIGHT$(A
$,14)
280 IF S>1 AND INT(S/8)=S/8 THEN
  GOSUB 420
290 M=495
300 PRINT@ 480,"S:";S;"M:";NM;
310 IF NM<1 AND M=495 THEN 340
320 IF M<32 THEN M=495
330 GOTO 60
340 IF S>R THEN R=S
350 PRINT@ 195,"SCORE :";S,"RECO
RD :";R;
360 PRINT@ 233,"UNE AUTRE ?";
370 R$=INKEY$
380 IF R$="" THEN 370
390 IF R$<>"N" THEN 50
400 CLS
410 END
420 A$=A1$
430 B$=B1$
440 NM=NM+B
450 FOR I=1 TO 300
460 NEXT I
470 S=S+B
480 RETURN
490 S=0
500 A$=""
```

```
510 B$=""
520 FOR I=1 TO 32
530 READ A,B
540 A$=A$+CHR$(A)
550 B$=B$+CHR$(B)
560 IF I/8=INT(I/8) THEN RESTORE
570 NEXT I
580 A1$=A$
590 B1$=B$
600 J$=CHR$(174)+CHR$(168)+CHR$(
172)
610 M=495
620 M$=CHR$(171)
630 NM=10
640 O$=CHR$(175)
650 CLS 3
660 PRINT@ 494,J$;
670 PRINT@ 0,A$;
680 PRINT@ 64,B$;
690 O1$=""
700 FOR I=1 TO 5
710 O1$=O1$+CHR$(175)
720 NEXT I
730 GOTO 60
740 DATA 175,175,164,172,172,172
,172,168,175,175,175,175,175,175
,175,175
```



TRACE

Deux joueurs s'affrontent pour se partager l'espace vital. Chacun doit s'efforcer, tout en se déplaçant, de ne jamais recouper sa trace ou celle de son adversaire, et de ne pas sortir du rectangle dessiné sur l'écran. Utilisez les joysticks ou les touches suivantes :

Joueur de droite : < P >, < L >, < ; > et < . >


Joueur de gauche : < W >, < A >, < S > et < Z >

```
10 REM *****
20 REM * TRACE *
30 REM *****
40 GOSUB 850
50 GOSUB 630
60 ON JK GOTO 150
70 FOR I=1 TO 50
80 NEXT I
90 D$=INKEY$
100 C1=(D$="L")-(D$=";" )+32*((D$
="P")-(D$="." ))
110 C2=(D$="A")-(D$="S")+32*((D$
="W")-(D$="Z" ))
120 IF C1<>0 THEN D1=C1
130 IF C2<>0 THEN D2=C2
140 GOTO 270
150 K0=JOYSTK(0)
160 K1=JOYSTK(1)
170 K2=JOYSTK(2)
180 K3=JOYSTK(3)
190 IF K0<26 AND K1>37 THEN S1=-
1
200 IF K0>37 AND K1<26 THEN S1=1
210 IF K0 <26 AND K1<26 THEN S1=
-32
```

```
220 IF K0>37 AND K1>37 THEN S1=3
2
230 IF K2<26 AND K3>37 THEN S2=-
1
240 IF K2>37 AND K3<26 THEN S2=1
250 IF K2<26 AND K3<26 THEN S2=-
32
260 IF K2>37 AND K3>37 THEN S2=3
2
270 IF S1<>0 THEN D1=S1
280 IF S2<>0 THEN D2=S2
290 P1=P1+D1
300 IF PEEK(P1)<>128 THEN 360
310 POKE P1,J1
320 P2=P2+D2
330 IF PEEK(P2)<>128 THEN 410
340 POKE P2,J2
350 GOTO 60
360 F2=F2+1
370 GOSUB 590
380 IF F2=10 THEN 460
390 D$=INKEY$
400 GOTO 50
410 F1=F1+1
420 GOSUB 590
430 IF F1=10 THEN 500
440 D$=INKEY$
450 GOTO 50
460 CLS
470 PRINT@ 165,"LE JOUEUR GAUCHE
GAGNE"
480 PRINT@ 202,F2;"A";F1
490 GOTO 530
500 CLS
510 PRINT@ 165,"LE JOUEUR DROIT
GAGNE"
520 PRINT@ 202,F1;"A";F2
530 R$=INKEY$
```



```
540 PRINT@ 266,"UNE AUTRE ?"  
550 R#=INKEY#  
560 IF R#="" THEN 540  
570 IF R#<>"N" THEN RUN  
580 END  
590 FOR I=5 TO 255 STEP 5  
600 SOUND I,1  
610 NEXT I  
620 RETURN  
630 CLS 0  
640 C=207  
650 J1=239  
660 J2=255  
670 P1=1272  
680 P2=1256  
690 D1=-1  
700 D2=1  
710 FOR I=1024 TO 1055  
720 POKE I,C  
730 POKE I+448,C  
740 NEXT I  
750 FOR I=1 TO 13  
760 POKE I*32+1024,C  
770 POKE I*32+1055,C  
780 NEXT I  
790 PRINT@ 480,"JOUEUR GAUCHE";F  
2,"JOUEUR DROIT";F1;  
800 POKE P1,J1  
810 POKE P2,J2  
820 S1=0  
830 S2=0  
840 RETURN  
850 CLS  
860 PRINT@ 170,"JOYSTICKS ?"  
870 D#=INKEY#  
880 IF D#="" THEN 870  
890 JK=- (D#="0")  
900 RETURN
```




SLALOM

Partez aux sports d'hiver sans risque de vous casser une jambe ! Lancez-vous du haut de la piste et essayez de passer le plus grand nombre possible de portes sans heurter les piquets. Frappez n'importe quelle touche pour changer de direction.

```
10 REM *****
20 REM * SLALOM *
30 REM *****
34 REM
35 REM INITIALISATION
36 REM
38 REM TABLEAU DES POSITIONS
39 REM DU SKIEUR
40 DIM S$(1)
50 FOR I=1 TO 32
60 E$=E$+CHR$(207)
70 NEXT I
79 REM SKIEUR ALLANT A GAUCHE
80 S$(0)=CHR$(201)
89 REM SKIEUR ALLANT A DROITE
90 S$(1)=CHR$(198)
99 REM FOND BLANC
100 CLS 5
```

```
108 REM DIRECTION INITIALE :
109 REM          GAUCHE
110 D=-1
118 REM POSITION INITIALE
119 REM DU SKIEUR
120 J=16
129 REM DESSIN DES PORTES
130 P#=CHR$(181)+CHR$(207)+CHR$(
207)+CHR$(170)
134 REM
135 REM BOUCLE PRINCIPALE
136 REM
140 FOR K=1 TO 300
145 REM CALCUL DES COORDONNEES
146 REM DU SKIEUR
150 Y=INT(J/32)*2
160 X=(J-16*Y)*2
165 REM SKIEUR AU NIVEAU D'UNE
166 REM PORTE?
170 IF K>=16 AND (K-5)/10=INT((K
-5)/10) THEN GOSUB 330
179 REM AFFICHAGE D'UNE PORTE?
180 IF K<284 AND K/10=INT(K/10)
THEN GOSUB 350
189 REM MOUVEMENT DU SKIEUR
190 IF INKEY#<>" " THEN D=-D
200 J=J+D
210 IF J<2 THEN J=2
220 IF J>29 THEN J=29
230 PRINT@ 511,E#;
240 PRINT@ J,S$(D/2+0.5);
250 NEXT K
254 REM
255 REM FIN DU PARCOURS
256 REM
260 PRINT@ 164,"PORTE(S) RATEE(S
) :";T;
```

```
270 PRINT@ 229,"UNE AUTRE DESCEN  
TE ?";  
280 D#=INKEY#  
290 IF D#="" THEN 280  
300 IF D#<>"N" THEN RUN  
310 CLS  
320 END  
324 REM  
325 REM PORTE RATEE?  
326 REM  
330 IF POINT(X-2,Y)<>0 OR POINT(  
X+4,Y)<>3 THEN IF POINT(X-4,Y)<>  
0 OR POINT(X+2,Y)<>3 THEN T=T+1:  
SOUND 1,1  
340 RETURN  
344 REM  
345 REM AFFICHAGE D'UNE PORTE  
346 REM  
350 P1=RND(3)-2  
360 P=P-6*P1  
370 IF P<482 THEN P=488  
380 IF P>506 THEN P=500  
390 PRINT@ P,P#;  
400 RETURN
```



RAMASSE-MIETTES


Voici une curieuse façon d'utiliser un ordinateur : vous devez vous efforcer de ramasser le plus vite possible les miettes qui jonchent la nappe. Vous disposez de 30 secondes pour un nettoyage complet. Les miettes sont représentées par des points noirs. Le joystick ou les touches <W>, <A>, <S> et <Z> vous permettent de contrôler votre ramasse-miettes (représenté par un carré noir).

```
10 REM *****
20 REM * RAMASSE-MIETTES *
30 REM *****
40 GOTO 120
50 K0=JOYSTK(0)
60 K1=JOYSTK(1)
70 IF K0<27 AND K1<27 THEN D=-32
80 IF K0>36 AND K1<27 THEN D=1
90 IF K0<27 AND K1>36 THEN D=-1
100 IF K0>36 AND K1>36 THEN D=32
110 GOTO 180
120 GOSUB 300
130 PRINT@ 480,"TEMPS :";T,"SCOR
E :";S;
140 IF T=0 THEN 640
150 ON JS GOTO 50
160 D#=INKEY#
```



```
170 D=(D$="A")-(D$="S")+32*( (D$="
"W")-(D$="Z"))
180 IF D<>0 THEN DO=D
190 D=0
200 P1=P+D0
210 IF PEEK(P1)=CB THEN P1=P
220 IF PEEK(P1)=CM THEN S=S+1:S0
UND 1,1:X=X+1
230 POKE P,CN
240 POKE P1,CJ
250 P=P1
260 T1=T1-0.2
270 T=INT(T1)
280 IF X=NM THEN 750
290 GOTO 130
300 CLS
310 PRINT@ 203,"JOYSTICK ?"
320 IF INKEY$<>"" THEN 320
330 D$=INKEY$
340 IF D$="" THEN 330
350 IF D$="0" THEN JS=1 ELSE JS=
0
360 CLS
370 CB=159
380 CM=110
390 CN=96
400 CJ=128
410 S=0
420 NM=10
430 X=0
440 FOR I=1024 TO 1055
450 POKE I,CB
460 POKE I+448,CB
470 NEXT I
480 FOR I=1 TO 13
490 POKE I*32+1024,CB
500 POKE I*32+1055,CB
510 NEXT I
```

```
520 FOR I=1 TO NM
530 P=RND(416)+1056
540 IF PEEK(P)<>96 THEN 530
550 POKE P,CM
560 NEXT I
570 P=RND(416)+1056
580 IF PEEK(P)<>96 THEN 570
590 POKE P,CJ
600 T1=30
610 T=T1
620 DO=0
630 RETURN
640 FOR I=1 TO 500
650 NEXT I
660 IF INKEY$<>" " THEN 660
670 IF S>R THEN R=S
680 PRINT@ 167,"RECORD :";R;
690 PRINT@ 231,"UNE AUTRE ?";
700 D$=INKEY$
710 IF D$="" THEN 700
720 IF D$<>"N" THEN 120
730 CLS
740 END
750 NM=NM+1
760 POKE P1,CN
770 GOSUB 430
780 GOTO 130
```




TANK

Essayez de suivre la route avec votre tank tout en évitant les mines qui s'y trouvent placées au hasard par l'ordinateur. Pour vous diriger, utilisez les touches <, > et <.>. Si le passage est bloqué, vous pouvez faire sauter une mine en tirant un de vos dix obus (frappez la barre d'espace).

```
10 REM *****
20 REM * TANK *
30 REM *****
40 GOSUB 210
44 REM
45 REM BOUCLE PRINCIPALE
46 REM
50 X$=INKEY$
60 T=T+(X$=G$)-(X$=D$)
70 POKE T,CR
80 PRINT R$
90 IF PEEK(T)<>CR THEN 490
100 POKE T,CT
110 IF X$=B$ THEN GOSUB 430
120 T1=T
130 PD=RND(7)+D
140 POKE PD,CM
150 PD=RND(7)+D
160 POKE PD,CM
170 S=S+1
```

```
180 FOR I=1 TO DL-S
190 NEXT I
194 REM
195 REM INITIALISATION
196 REM
200 GOTO 50
210 CLS
220 T=1232
230 T1=T
240 S=0
250 CT=84
260 CR=96
270 CM=106
280 DL=200
290 D=1484
300 R$=""
310 FOR I=1 TO 11
320 R$=R$+" "
330 NEXT I
340 R$=R$+"XX"      XX"
350 D$="."
360 G$=","
370 B$=" "
380 OB=10
390 FOR I=1 TO 15
400 PRINT R$
410 NEXT I
420 RETURN
424 REM
425 REM TIR
426 REM
430 IF OB=0 THEN 480
440 OB=OB-1
450 FOR I=1 TO 5
460 POKE T+32,CR
470 NEXT I
480 RETURN
484 REM
```

```
485 REM PERDU
486 REM
490 FOR I=1 TO 10
500 POKE T,CR
510 FOR J=1 TO 100
520 NEXT J
530 POKE T,191
540 FOR J=1 TO 100
550 NEXT J
560 NEXT I
570 IF S>R THEN R=S
580 CLS
590 PRINT@ 234,"SCORE :";S;
600 PRINT@ 298,"RECORD :";R;
610 PRINT@ 362,"UNE AUTRE ?";
620 X$=INKEY$
630 IF X$<>" " THEN 620
640 X$=INKEY$
650 IF X$="" THEN 640
660 IF X$<>"N" THEN 40
670 CLS
680 END
```





ALPHABET

Voici un jeu d'adresse assez difficile. Vous tenterez de marquer le plus grand nombre de points possibles en effaçant les lettres affichées par l'ordinateur. Utilisez le joystick ou les touches <W> (haut), <Z> (bas), <A> (gauche) et <S> (droite). *Attention* : vous devez effacer les lettres dans l'ordre alphabétique en évitant les obstacles placés au hasard sur l'écran, et ceci en un temps limité. (Le temps qui vous reste est affiché au bas de l'écran.) Lorsque toutes les lettres sont effacées, le jeu reprend avec une lettre supplémentaire.

```
10 REM *****
20 REM * ALPHABET *
30 REM *****
40 GOSUB 790
50 S=0
60 X=0
70 CN=223
80 CP=191
90 GOSUB 520
100 FOR I=1 TO X
110 ON JK GOTO 160
120 D#=INKEY#
130 D=(D#="A")-(D#="S")+32*((D#="W")-(D#="Z"))
140 IF D<>0 THEN DO=D
150 GOTO 240
160 K0=JOYSTK(0)
170 K1=JOYSTK(1)
180 IF K0<27 AND K1<27 THEN KS=-
32
```

```
190 IF K0>36 AND K1<27 THEN KS=1
200 IF K0<27 AND K1>36 THEN KS=-
1
210 IF K0>27 AND K1>27 THEN KS=3
2
220 IF KS<>0 THEN D0=KS
230 KS=0
240 T=T-0.1
250 PRINT@ 480,"TEMPS : ";INT(T+1
);
260 IF T<0 THEN 410
270 P=P+D0
280 C=PEEK(P)
290 IF C=I+64 THEN S=S+I: SOUND 1
,1:GOTO 350
300 IF C<>223 THEN P=P1
310 POKE P1,CN
320 POKE P,CP
330 P1=P
340 GOTO 110
350 POKE P1,CN
360 POKE P,CP
370 P1=P
380 NEXT I
390 GOSUB 520
400 GOTO 100
410 D$=INKEY$
420 IF R<S THEN R=S
430 PRINT@ 166,"TEMPS ECOULE";
440 PRINT@ 234,"SCORE : ";S;
450 PRINT@ 266,"RECORD : ";R;
460 PRINT@ 326,"UNE AUTRE ?";
470 D$=INKEY$
480 IF D$="" THEN 470
490 IF D$<>"N" THEN 50
500 CLS
510 END
520 CLS 6
```

```
530 X=X+1
540 FOR I=1024 TO 1055
550 POKE I,159
560 POKE 448+I,159
570 NEXT I
580 FOR I=1 TO 13
590 POKE I*32+1024,159
600 POKE I*32+1055,159
610 NEXT I
620 FOR I=1 TO 70
630 GOSUB 760
640 POKE P,128
650 NEXT I
660 FOR I=1 TO X
670 GOSUB 760
680 POKE P,I+64
690 NEXT I
700 GOSUB 760
710 POKE P,CP
720 P1=P
730 T=50
740 DO=0
750 RETURN
760 P=RND(414)+1056
770 IF PEEK(P)<>CN THEN 760
780 RETURN
790 CLS
800 PRINT@ 203,"JOYSTICK ?"
810 D$=INKEY$
820 IF D$="" THEN 810
830 IF D$="0" THEN JK=1
840 RETURN
```





ATTERRISSAGE

Après un long voyage en apesanteur, poser une navette spatiale en douceur n'est pas chose aisée ; mais grâce à votre ordinateur, vous allez être en mesure de vous entraîner sans danger. Vous devez poser votre navette sur l'aire prévue à cet effet. Vous pouvez vous diriger vers la droite et vers la gauche à l'aide des touches <, > et <.> ou freiner votre descente avec la barre d'espacement. La quantité de carburant disponible est indiquée par la longueur du trait horizontal, en haut de l'écran.

```
10 REM *****
20 REM * ATERRISSAGE *
30 REM *****
40 PCLEAR 4
50 DIM N(15,20)
60 DIM R(15,20)
70 DIM C(15,10)
80 FU=100
90 GOTO 560
100 GOSUB 710
110 DV=DV+1
120 IF FU<1 THEN D$="":GOTO 140
130 D$=INKEY$
140 DV=DV+3*(D$=" ")
150 DH=DH+(D$=".">)-(D$=",">)
```

```
160 IF D#<>" THEN FU=FU-1:FF=1
170 DRAW "COL"+STR$(FF)
180 FF=0
190 IF MH<0 OR NH>255 THEN 220
200 IF MV<0 THEN 220
210 PUT (MH,MV)-(NH,NV),R,PSET
220 NH=NH+DH
230 NV=NV+DV
240 IF NV>187 THEN 310
250 MV=NV-20
260 MH=NH-15
270 IF MH<0 OR NH>255 THEN 110
280 IF MV<0 THEN 110
290 PUT (MH,MV)-(NH,NV),N,PSET
300 GOTO 110
310 IF DV>5 THEN 390
320 IF ABS(NH-A)>4 THEN 390
330 IF ABS(DH)>1 THEN 390
340 PUT (MH,167)-(NH,187),N,PSET
350 FOR I=0 TO 1000
360 NEXT I
370 S=S+10
380 GOTO 100
390 IF MH<0 OR NH>240 THEN 430
400 IF MV<0 OR NV>191 THEN 420
410 PUT (MH,MV)-(NH,NV),R,PSET
420 PUT (MH,181)-(NH,191),C,PSET
430 FOR I=1 TO 1000
440 NEXT I
450 CLS
460 SCREEN 0,0
470 PRINT@ 162,"VOTRE NAVETTE S'
EST ECRASEE"
480 PRINT@ 234,"SCORE :";S
490 PRINT@ 298,"UNE AUTRE ?"
500 IF INKEY#<>" THEN 500
510 D#=INKEY#
520 IF D#="" THEN 510
```

```
530 IF D$<>"N" THEN RUN
540 CLS
550 END
560 PCLS
570 CLS
580 PRINT@ 166,"ATTENTION..."
590 PMODE 4
600 FOR I=0 TO 15
610 FOR J=0 TO 10
620 C(I,J)=RND(500)
630 NEXT J
640 NEXT I
650 PRINT@ 166,"TENEZ-VOUS PRET"
660 DRAW "BM50,50R2L1U3E3H4U6E4R
7F4D6G4F3D3L1R2L1U3H3L3D2R1D1U1L
3D1U1R1U2L3"
670 PAINT (55,40)
680 GET (50,30)-(65,50),N,G
690 GET (0,0)-(15,20),R,G
700 GOTO 100
710 PCLS
720 PMODE 4
730 A=RND(216)+20
740 DRAW "BM"+STR$(A)+"",188R4L23
D1R23D1L23"
750 NH=120
760 NV=20
770 DH=0
780 DV=0
790 MV=0
800 MH=NH-15
810 FU=FU+10
820 DRAW "BMO,3R"+STR$(FU)
830 SCREEN 1,0
840 RETURN
```

CHASSE AU CANARD

Votre micro-ordinateur vous invite à une partie de chasse au canard. Les canards volent de droite à gauche en haut de l'écran. Vous vous déplacez en utilisant les touches <A> et <S>. La touche <W> vous permet de vous arrêter. Vous pouvez tirer autant de cartouches que vous voulez, mais il ne passera que 20 canards, et votre but est d'en abattre le plus possible. Pour tirer, frappez la barre d'espace. Si vous atteignez un canard, vous marquez un point et vous le verrez tomber en battant des ailes.


Dans la version présentée, le jeu est assez facile. Si vous désirez augmenter la difficulté, remplacez la ligne 290 par :

```
290 IF T=C-1 THEN 410
```

```
10 REM *****
20 REM * CHASSE AU CANARD *
30 REM *****
40 GOSUB 640
50 X=-X
60 Y=X+0.5
70 C=C-0.5
80 IF C<0 THEN GOSUB 580
90 IF NC<1 THEN 340
100 PRINT@ C,CC$(INT(Y));
110 PJ=FJ+D1
120 IF PJ<480 THEN PJ=480
130 IF PJ>508 THEN PJ=508
140 PRINT@ PJ,J$;
```

```
150 IF ABS(T-PJ)>1 THEN T=T-32:G
OTO 170
160 T=PJ
170 IF T<32 THEN GOSUB 290
180 PRINT@ T,J#;
190 IF T<>PJ THEN PRINT@ T+32,N#
;
200 R#=INKEY#
210 D=(R#="A")-(R#="S")
220 IF D<>0 THEN D1=D
230 IF R#="W" THEN D1=0:GOTO 50
240 IF R#<>" " OR T<>PJ THEN 50
250 SOUND 1,1
260 IF T<>PJ THEN PRINT@ T,N#;
270 T=PJ-32
280 GOTO 50
290 IF ABS(T-C+1)<=1 THEN 410
300 PRINT@ T+32,N#;
310 NT=NT-1
320 T=PJ
330 IF NC>0 THEN RETURN
340 PRINT@ 166,"SCORE :";S;
350 PRINT@ 230,"UNE AUTRE ?";
360 R#=INKEY#
370 IF R#="" THEN 360
380 IF R#<>"N" THEN RUN
390 CLS
400 END
410 SOUND 255,2
420 S=S+1
430 D=C
440 C=29
450 FOR I=15 TO 1 STEP-1
460 X=-X
470 Y=X+0.5
480 PRINT@ D,N#;
490 SOUND 15*I,1
500 PRINT@ D+32,CC#(INT(Y));
```

```
510 D=D+32
520 NEXT I
530 FOR I=1 TO 200
540 NEXT I
550 CLSO
560 NC=NC-1
570 GOTO 50
580 PRINT@ O,N#;
590 FOR I=1 TO 200
600 NEXT I
610 C=29
620 NC=NC-1
630 RETURN
640 DIM CC$(1)
650 N#=CHR$(128)+CHR$(128)
660 J#=CHR$(128)+CHR$(132)+CHR$(
128)
670 CC$(0)=CHR$(134)+CHR$(130)+C
HR$(128)
680 CC$(1)=CHR$(137)+CHR$(136)+C
HR$(128)
690 PJ=495
700 NC=20
710 T=PJ
720 X=0.5
730 C=29
740 CLS 0
750 RETURN
```




SERPENT

Dans ce jeu, vous êtes un serpent qui se déplace en ondulant sur l'écran. Le changement de direction s'effectue en tapant n'importe quelle touche. Pour pouvoir vous déplacer, vous devez vous nourrir. Heureusement, vous êtes entouré par un grand nombre de champignons. Mais attention ! Si les bleus sont excellents, vous devez absolument éviter les rouges qui, eux, sont vénéneux. Chaque champignon bleu vous apporte suffisamment de calories pour avancer de dix lignes. Essayez de ne pas mourir de faim sans pour autant finir empoisonné !

```
10 REM *****
20 REM * SERPENT *
30 REM *****
38 REM T=POSITION INITIALE
39 REM DU SERPENT
40 T=239
49 REM H=SCORE
50 H=0
59 REM R#=CHAMPIGNON ROUGE
60 R#=CHR$(191)
69 REM B#=CHAMPIGNON BLEU
70 B#=CHR$(175)
79 REM T#=CARACTERE SERPENT
80 T#="S"
89 REM S=ENERGIE
90 S=100
```

```
100 CLS
109 REM D= DIRECTION (1 OU -1)
110 D=1
114 REM
115 REM BOUCLE PRINCIPALE
116 REM
119 REM CHANGEMENT DE DIRECTION
120 D#=INKEY#
130 IF D#<>" " THEN D=-D
140 T=T+D
150 IF T<225 THEN T=226
160 IF T>254 THEN T=253
170 T1=T+32
179 REM CALCUL COORDONNEES
180 Y=INT(T1/32)*2
190 X=(T1-16*Y)*2
199 REM CHAMPIGNON ROUGE?
200 IF POINT(X,Y)=4 THEN 300
209 REM CHAMPIGNON BLEU?
210 IF POINT(X,Y)=3 THEN S=S+10:
H=H+10:SOUND 1,1
220 PRINT@ 511," ";
229 REM AFFICHAGE CHAMPIGNONS
230 PRINT@ 480+RND(30),B#;
240 IF RND(2)=1 THEN PRINT@ 480+
RND(30),R#;
249 REM AFFICHAGE SERPENT
250 PRINT@ T,T#;
259 REM COMPTE ENERGIE
260 S=S-1
269 REM MORT DE FAIM?
270 IF S=0 THEN 300
279 REM INCREMENTATION DU SCORE
280 H=H+1
290 GOTO 120
294 REM
295 REM FIN
296 REM
```

```
300 SOUND 1,1
310 PRINT@ 511," ";
320 PRINT@ T,T#;
330 FOR I=1 TO 5
340 SOUND 1,1
350 FOR J=1 TO 50
360 D#=INKEY#
370 NEXT J
380 NEXT I
389 REM RECORD BATTU?
390 IF H>R THEN R=H
400 PRINT@ 66,"SCORE :";H,"RECOR
D :";R;
410 PRINT@ 138,"UNE AUTRE ?";
420 D#=INKEY#
430 IF D#="" THEN 420
440 IF D#<>"N" THEN 40
450 CLS
460 END
```

SIMON

Dans cette version du jeu bien connu, vous utiliserez les touches <1>, <2>, <3> et <4> pour répéter la séquence produite de façon aléatoire par l'ordinateur. A chaque touche correspondent une couleur et un son :

(du plus grave au plus aigu)


- 1 : blanc
- 2 : bleu
- 3 : violet
- 4 : orange

La limite de la longueur de la séquence à répéter est 100. Lorsque vous faites une erreur, le jeu s'arrête et votre score, correspondant à la longueur de la plus longue séquence que vous avez réussi à reproduire, est affiché ainsi que le record du jour.

```
10 REM *****
20 REM * SIMON *
30 REM *****
40 DIM T(100)
50 X=0
60 CLS 0
70 GOSUB 370
79 REM X=LONGUEUR DE LA SEQUENCE
80 X=X+1
90 T(X)=RND(4)
100 FOR I=1 TO X
```

```
110 GOSUB 330
120 NEXT I
130 FOR I=1 TO X
139 REM ATTENTE
140 FOR J=1 TO 1000
150 D$=INKEY$
160 IF D$<>" " THEN 190
170 NEXT J
179 REM TROP LONG, PERDU
180 GOTO 250
189 REM ERREUR?
190 IF ASC(D$)-48<>T(I) THEN 250
200 GOSUB 330
210 NEXT I
220 FOR I=1 TO 500
230 NEXT I
240 GOTO 80
244 REM
245 REM PERDU
246 REM
250 X=X-1
260 IF X>R THEN R=X
270 PRINT@ 324,"SCORE :";X,"RECO
RD :";R;
280 PRINT@ 394,"UNE AUTRE ?";
290 D$=INKEY$
300 IF D$="" THEN 290
310 IF D$<>"N" THEN 50
320 END
324 REM
325 REM AFFICHAGE DE LA SEQUENCE
326 REM
330 POKE 1285+4*T(I),191+T(I)*16
340 SOUND 50+T(I)*45,2
350 POKE 1285+4*T(I),128
360 RETURN
364 REM
365 REM INITIALISATION
```

```
366 REM
369 REM TRACE DU CADRE
370 FOR I=1024 TO 1055
380 POKE I,255
390 POKE I+448,255
400 NEXT I
410 FOR I=1 TO 13
420 POKE 1024+I*32,255
430 POKE 1055+I*32,255
440 NEXT I
449 REM AFFICHAGE DES CHIFFRES
450 FOR I=1 TO 4
460 POKE 1221+4*I,I+48
470 NEXT I
480 FOR I=1 TO 7
490 POKE 1099+I,239
500 POKE 1163+I,239
510 NEXT I
520 POKE 1132,239
530 POKE 1138,239
540 PRINT@ 109,"SIMON";
550 RETURN
```



CRABES

Vous devez maintenant aider une pauvre crevette à regagner la mer en évitant les crabes voraces qui patrouillent sur la plage. Chaque crevette amenée au but rapporte 1 point. Vous disposez de cinq vies pour tenter de marquer un score maximum. Utilisez les touches <W> pour avancer et <Z> pour reculer.


```
10 REM *****
20 REM * CRABES *
30 REM *****
40 GOSUB 570
50 PRINT@ 384,"VIE(S) REST.",NF
60 A$=RIGHT$(A$,1)+LEFT$(A$,31)
70 B$=RIGHT$(B$,31)+LEFT$(B$,1)
80 PRINT@ X1,B$;
90 PRINT@ X2,A$;
100 PRINT@ X3,B$;
110 PRINT@ X4,A$;
120 D$=INKEY$
130 P=P+32*((D$="W")-(D$="Z"))
```


```
140 IF P>1391 THEN P=1391
150 IF P=1135 THEN 240
160 C=PEEK(P)
170 IF C<>CN AND C<>CF THEN 360
180 POKE P1,CN
190 POKE P,CF
200 P1=P
210 T=T+1
220 IF T>500 THEN 450
230 GOTO 50
240 POKE P1,CN
250 POKE P,CF
260 SOUND 1,1
270 FOR I=1 TO 200
280 NEXT I
290 POKE P,CN
300 P=1391
310 P1=P
320 S=S+1
330 PRINT@ 0,"SCORE :";S,"RECORD
   :";R
340 GOSUB 780
350 GOTO 50
360 NP=NP-1
370 POKE P1,CN
380 POKE P,128
390 GOSUB 810
400 IF NP=0 THEN 450
410 P=1391
420 P1=P
430 GOSUB 780
440 GOTO 50
450 CLS
460 IF S>R THEN R=S
470 IF T>500 THEN PRINT@ 170,"TE
MPS ECOULE";
480 PRINT@ 260,"SCORE :";S,"RECO
RD :";R
```

```
490 PRINT@ 330,"UNE AUTRE ?";
500 D#=INKEY#
510 IF D#<>" " THEN 500
520 D#=INKEY#
530 IF D#="" THEN 520
540 IF D#<>"N" THEN 40
550 CLS
560 END
570 CLS 2
580 A#=""
590 B#=""
600 S=0
610 NP=5
620 P=1391
630 P1=P
640 X1=128
650 X2=160
660 X3=224
670 X4=256
680 T=0
690 CP=191
700 CN=159
710 RESTORE
720 FOR I=1 TO 32
730 READ A,B
740 A#=A#+CHR$(A)
750 B#=B#+CHR$(B)
760 NEXT I
770 POKE P,CP
780 X=RND(31)
790 A#=RIGHT$(A#,X)+LEFT$(A#,32-
X)
800 RETURN
810 SOUND 15,12
820 SOUND 15,9
830 SOUND 15,3
840 SOUND 15,12
850 SOUND 55,9
```



```
860 SOUND 45,3
870 SOUND 45,9
880 SOUND 15,3
890 SOUND 15,9
900 SOUND 1,3
910 SOUND 15,12
920 D$=INKEY$
930 IF D$<>" " THEN 910
940 RETURN
950 DATA 159,159,153,153,150,150
,159,159,159,159,159,153,159,150
960 DATA 153,159,150,159,159,159
,159,159,159,159,159,153,159,150
970 DATA 159,159,159,159,153,153
,150,150,159,159,159,153,159,150
980 DATA 153,159,150,159,159,153
,153,150,150,159,159,159,159,159
990 DATA 159,159,153,159,150,153
,159,150
```






GRAND-PRIX

Au volant de votre formule 1, essayez de parcourir la plus grande distance possible. Votre voiture dispose de deux vitesses que vous pouvez sélectionner à l'aide des touches <1> et <2>. La direction est commandée par les touches <, > et <.>. En seconde vitesse, votre voiture roule deux fois plus vite. Mais gare à l'accident !

```
10 REM *****
20 REM * GRAND-PRIX *
30 REM *****
40 GOSUB 390
44 REM
45 REM BOUCLE PRINCIPALE
46 REM
50 D$=INKEY$
59 REM CHANGEMENT DE VITESSE
60 IF D$="1" THEN T=1
70 IF D$="2" THEN T=2
79 REM DIRECTION
80 V=V+(D$=",")-(D$=".">)
90 P=V+32
99 REM ACCIDENT?
100 IF PEEK(P)<>128 THEN 240
```

```
110 R=R+(RND(2)=1)-(RND(2)=1)
120 IF R<RN THEN R=RN
130 IF R>RM THEN R=RM
139 REM DEFILEMENT ECRAN
140 PRINT@ PF,C#;
150 POKE V1-32,CN
159 REM AFFICHAGE ROUTE
160 PRINT@ R,R#;
169 REM AFFICHAGE VOITURE
170 POKE V,CV
179 REM COMPTE KILOMETRE
180 K=K+T
190 DL=(2-T)*50
199 REM DELAI
200 FOR I=1 TO DL
210 NEXT I
220 V1=V
230 GOTO 50
234 REM
235 REM ACCIDENT
236 REM
240 POKE P,191
250 POKE V1,CN
260 FOR I=1 TO 4
270 FOR J=1 TO 20
280 SOUND J*10,1
290 NEXT J
300 NEXT I
310 PRINT@ 166,"KMS PARCOURUS : "
;K;
320 D#=INKEY#
330 PRINT@ 233,"UNE AUTRE ?";
340 D#=INKEY#
350 IF D#="" THEN 340
360 IF D#<>"N" THEN RUN
370 CLS
380 END
384 REM
```

```
385 REM INITIALISATION
386 REM
390 CLS
400 CN=128
409 REM R#=ROUTE
410 R#=CHR$(159)+CHR$(128)+CHR$(
128)+CHR$(128)+CHR$(159)
420 R=13
430 T=1
440 CV=73
450 V=1231
460 V1=V
469 REM AFFICHAGE ROUTE
470 FOR I=0 TO 15
480 PRINT@ R,R#;
490 R=R+32
500 NEXT I
510 R=R-32
519 REM AFFICHAGE VOITURE
520 POKE 1024+V,CV
530 RM=506
540 RN=480
550 C#=" "
560 PF=511
570 RETURN
```



CASSE-BRIQUES


Voici un jeu que l'on a l'habitude de voir dans les cafés. Grâce à votre ordinateur, vous allez maintenant pouvoir en profiter sans dépenser un centime. Le but du jeu est simple : essayer de détruire un mur de briques à l'aide d'une balle que vous devez renvoyer avec votre raquette. Chaque brique cassée rapporte un point. Lorsque le mur est entièrement détruit, un nouveau mur apparaît. Vous disposez de dix balles pour tenter de marquer un score maximum. Utilisez le joystick ou les touches <A>, <S> et la barre d'espace pour déplacer la raquette.

```
10 REM *****
20 REM * CASSE-BRIQUES *
30 REM *****
40 S=0:NB=0
50 GOSUB 530
60 V=V+DV:H=H+DH
70 POKE B+1024,CN
80 B=V*32+H
90 L=PEEK(B+1024)
100 IF L<>223 AND L<>128 THEN DV
=-DV:K=0:S=S+1
110 POKE B+1024,CB
120 IF V=13 AND ABS(R-29-B)>1 TH
EN 320
130 IF V=13 AND H>2 AND H<29 THE
N POKE B+1024,CN:H=H+CH
```



```
140 IF V=13 OR V=1 THEN DV=-DV
150 IF H=1 OR H=30 THEN DH=-DH
160 ON JS GOSUB 220,250
170 IF R<446 THEN R=446
180 IF R>475 THEN R=475
190 PRINT@ R,R$;
200 IF S/120=INT(S/120) AND K=0
THEN GOSUB 670
210 GOTO 60
220 L=INT(JOYSTK(0)/2+445)
230 IF ABS(L-R)>1 THEN R=R+2*SGN
(L-R):CH=SGN(L-R)
240 RETURN
250 D$=INKEY$
260 D=2*((D$="A")-(D$="S"))
270 IF D$=" " THEN DO=0
280 IF D<>0 THEN DO=D
290 R=R+DO
300 CH=SGN(DO)
310 RETURN
320 NB=NB+1
330 FOR I=1 TO 5
340 SOUND 1,1
350 FOR J=1 TO 50
360 NEXT J,I
370 IF NB=11 THEN 410
380 POKE B+1024,CN
390 GOSUB 760
400 GOTO 60
410 IF S>R1 THEN R1=S
420 PRINT@ 166,"SCORE :";S;
430 PRINT@ 230,"RECORD :";R1;
440 PRINT@ 294,"UNE AUTRE ?";
450 FOR I=1 TO 100
460 D$=INKEY$
470 NEXT I
480 D$=INKEY$
490 IF D$="" THEN 480
```

```
500 IF D$<>"N" THEN 40
510 CLS
520 END
530 CLS
540 PRINT@ 203,"JOYSTICK ?";
550 D$=INKEY$
560 IF D$="" THEN 550
570 IF D$="0" THEN JS=1 ELSE JS=
2
580 CLS 6
590 R=461
600 FOR I=1024 TO 1055
610 POKE I,128
620 NEXT I
630 FOR I=1 TO 12
640 POKE I*32+1024,128
650 POKE I*32+1055,128
660 NEXT I
670 FOR I=1057 TO 1086
680 POKE I+32,159
690 POKE I+64,239
700 POKE I+96,255
710 POKE I+128,175
720 NEXT I
730 K=1
740 R$=CHR$(223)+CHR$(223)+CHR$(
211)+CHR$(211)+CHR$(211)+CHR$(22
3)+CHR$(223)
750 CB=79:CN=223
760 V=13:DV=-1
770 H=RND(28)+1
780 B=V*32+H
790 DH=(RND(2)-1.5)*2
800 B1=B:DO=0
810 RETURN
```



POURSUITE

Le voleur s'est échappé, emportant le magot. (Il est représenté par un carré noir.) Il se cache dans la ville et vous avez trente minutes pour le débusquer et l'arrêter. Attention, pas de précipitation ! En effet, si vous vous jetez sur lui sans réfléchir, il y a toutes les chances pour qu'il vous file entre les doigts. La meilleure façon de vous y prendre est de l'aborder de côté. (Efficace à tous les coups à condition de ne pas le rater !) Si vous ne vous sentez pas assez sûr de vous, attaquez-le de face, ce qui est plus facile mais moins efficace car moins discret. Encore un conseil : n'essayez pas de le poursuivre ; cela ne vous mènerait à rien car il est aussi rapide que vous. Observez plutôt ses mouvements comme un vrai détective. Quand vous le verrez tourner en rond, approchez-vous sans faire de bruit et surgissez au bon moment. Mais rappelez-vous, le temps presse !

Pour vous déplacer, utilisez le joystick ou les touches suivantes :


- < W > : haut
- < A > : gauche
- < S > : droite
- < Z > : bas


```
10 REM *****
20 REM * POURSUITE *
30 REM *****
40 GOSUB 980
50 S=0
60 CN=159
70 CV=128
```

```
80 CP=191
90 GOSUB 640
100 ON JK GOTO 150
110 D#=INKEY#
120 D=(D#="A")-(D#="S")+32*((D#="W")-(D#="Z"))
130 IF D<>0 THEN D0=D
140 GOTO 230
150 K0=JOYSTK(0)
160 K1=JOYSTK(1)
170 IF K0<27 AND K1<27 THEN KS=-32
180 IF K0>36 AND K1<27 THEN KS=1
190 IF K0<27 AND K1>36 THEN KS=-1
200 IF K0>27 AND K1>27 THEN KS=32
210 IF KS<>0 THEN D0=KS
220 KS=0
230 T=T-0.1
240 PRINT@ 480,"TEMPS :";INT(T+1);
250 IF T<0 THEN 400
260 P=P+D0
270 C=PEEK(P)
280 IF C=128 THEN 920
290 IF C<>159 THEN P=P1
300 POKE P1,CN
310 POKE P,CP
320 P1=P
330 V=V+DV
340 IF PEEK(V)<>CN THEN GOSUB 500
350 IF PEEK(V)<>CN THEN 330
360 POKE V1,CN
370 POKE V,CV
380 V1=V
390 GOTO 100
```

```
400 D$=INKEY$
410 IF R<S THEN R=S
420 PRINT@ 166,"TEMPS ECOULE";
430 PRINT@ 234,"SCORE :";S;
440 PRINT@ 266,"RECORD :";R;
450 PRINT@ 326,"UNE AUTRE ?";
460 D$=INKEY$
470 IF D$="" THEN 460
480 IF D$<>"N" THEN 50
490 END
500 D2=D2+1
510 GOSUB 600
520 IF PEEK(V1+DV)=CN THEN V=V1+
DV:RETURN
530 D2=D2-2
540 GOSUB 600
550 IF PEEK(V1+DV)=CN THEN V=V1+
DV:RETURN
560 D2=D2-1
570 GOSUB 600
580 V=V1+DV
590 RETURN
600 IF D2>4 THEN D2=D2-4
610 IF D2<1 THEN D2=D2+4
620 DV=(D2=1)-(D2=3)+32*((D2=2)-
(D2=4))
630 RETURN
640 CLS 2
650 FOR I=1024 TO 1055
660 POKE I,175
670 POKE 448+I,175
680 NEXT I
690 FOR I=1 TO 13
700 POKE I*32+1024,175
710 POKE I*32+1055,175
720 NEXT I
730 FOR I=1 TO 70
740 GOSUB 890
```

```
750 POKE P,96
760 NEXT I
770 GOSUB 890
780 V=P
790 POKE V,CV
800 V1=V
810 GOSUB 890
820 POKE P,CP
830 P1=P
840 T=30
850 D0=0
860 DV=0
870 D2=0
880 RETURN
890 P=RND(414)+1056
900 IF PEEK(P)<>CN THEN 890
910 RETURN
920 FOR I=1 TO 5
930 SOUND 35,10
940 SOUND 5,10
950 NEXT I
960 S=S+1
970 GOTO 90
980 CLS
990 PRINT@ 203,"JOYSTICK ?"
1000 D$=INKEY$
1010 IF D$="" THEN 1000
1020 IF D$="0" THEN JK=1
1030 RETURN
```





ROBOTS

Vous voici seul, abandonné sur une planète défendue par des robots meurtriers. Le sol est truffé de mines qu'il vous faut absolument éviter. Les mines sont représentées sur l'écran par des "X". Au début du jeu, cinq robots (représentés par des losanges) sont présents sur le terrain. Sans perdre une seconde, ils se précipitent sur vous en suivant toujours le plus court chemin. Par chance, les robots sont aveugles et ne voient pas les mines qui se trouvent entre eux et vous, ce qui vous permet, en vous déplaçant judicieusement, de les éliminer. Utilisez pour cela le joystick ou les touches :

< Q > < W > < E >
< A > < S > < D >
< Z > < X > < C >


suivant la direction que vous avez choisie. La touche < S > vous servira à vous arrêter. Lorsque tous les robots sont éliminés, le jeu reprend avec un robot supplémentaire. Si vous sautez sur une mine ou si un robot vous tue, tout n'est pas perdu. Vous disposez en effet de cinq vies. Si vous désirez changer le nombre de mines, modifiez la valeur de la variable NM à la ligne 60.


```
10 REM *****  
20 REM * ROBOTS *  
30 REM *****  
40 NH=5  
50 N1=5  
60 NM=40  
70 NR=N1  
80 DIM R(30)
```

```
90 GOSUB 1130
100 GOSUB 740
110 ON JS GOSUB 570,680
120 X=X+D
130 C=PEEK(1024+X)
140 IF C<>96 AND C<>72 THEN 430
150 PRINT@ X1," ";
160 PRINT@ X,"H";
170 X1=X
180 XY=INT(X/32)
190 XX=X-XY*32
200 T=0
210 FOR I=1 TO NR
220 IF R(I)=0 THEN 350
230 T=1
240 RY=INT(R(I)/32)
250 RX=R(I)-RY*32
260 R1=RY+SGN(XY-RY)
270 R2=RX+SGN(XX-RX)
280 RR=32*R1+R2
290 C=PEEK(1024+RR)
300 IF C=88 OR C=82 THEN S=S+1:P
RINT@ R(I)," ";R(I)=0:GOTO 350
310 IF C=72 THEN 430
320 PRINT@ R(I)," ";
330 PRINT@ RR,"R";
340 R(I)=RR
350 NEXT I
360 IF T=0 THEN 400
370 FOR I=1 TO 500
380 NEXT I
390 GOTO 110
400 S=S+10
410 IF NR<30 THEN NR=NR+1
420 GOTO 100
430 NH=NH-1
440 PRINT@ X,CHR$(239);
450 FOR I=1 TO 20
```

```
460 SOUND 100,1
470 NEXT I
480 IF NH>0 THEN NR=N1:GOTO 100
490 CLS
500 PRINT@ 166,"SCORE :";S;
510 PRINT@ 230,"UNE AUTRE ?";
520 IF INKEY#<>" " THEN 520
530 D#=INKEY#
540 IF D#="" THEN 530
550 IF D#<>"N" THEN RUN
560 END
570 D#=INKEY#
580 IF D#="Q" THEN D=-33
590 IF D#="W" THEN D=-32
600 IF D#="E" THEN D=-31
610 IF D#="A" THEN D=-1
620 IF D#="S" THEN D=0
630 IF D#="D" THEN D=1
640 IF D#="Z" THEN D=31
650 IF D#="X" THEN D=32
660 IF D#="C" THEN D=33
670 RETURN
680 D=0
690 D=D+(JOYSTK(0)=0)
700 D=D-(JOYSTK(0)=63)
710 D=D+32*(JOYSTK(1)=0)
720 D=D-32*(JOYSTK(1)=63)
730 RETURN
740 CLS
750 D=0
760 PRINT@ 480,"SCORE :";S;
770 IF NH=1 THEN 810
780 FOR I=1 TO NH-1
790 PRINT "H";
800 NEXT I
810 PRINT@ 479,"X";
820 PRINT@ 0,"X";
830 FOR I=1 TO 30
```

```
840 PRINT@ I,"X";
850 PRINT@ 448+I,"X";
860 NEXT I
870 FOR I=0 TO 13
880 PRINT@ I*32+31,"XX";
890 NEXT I
900 FOR I=1 TO NM
910 P=RND(416)+31
920 IF PEEK(1024+P)<>96 THEN 910
930 PRINT@ P,"X";
940 NEXT I
950 FOR I=1 TO NR
960 R(I)=RND(416)+31
970 IF PEEK(1024+R(I))<>96 THEN
980 PRINT@ R(I),"R";
990 NEXT I
1000 X=RND(416)+31
1010 IF PEEK(1024+X)<>96 THEN 10
00
1020 X1=X
1030 FOR I=1 TO 5
1040 PRINT@ X,CHR$(239);
1050 SOUND 1,1
1060 FOR J=1 TO 50
1070 NEXT J
1080 PRINT@ X,"H";
1090 FOR J=1 TO 50
1100 NEXT J
1110 NEXT I
1120 RETURN
1130 CLS
1140 PRINT@ 203,"JOYSTICK ?"
1150 D$=INKEY$
1160 IF D$="" THEN 1150
1170 IF D$="0" THEN JS=2 ELSE JS
=1
1180 RETURN
```





PARACHUTE

Essayez, en sautant d'un hélicoptère en vol, d'atteindre la cible qui se trouve au sol. Une première pression sur une touche vous permet de descendre verticalement en chute libre. Une deuxième pression entraîne l'ouverture du parachute. La descente continue alors plus lentement et avec un angle de 45 degrés car le vent vous pousse. Plus vous attendez pour ouvrir votre parachute et moins vous serez déporté. Mais n'attendez pas trop longtemps car au-dessous de cent mètres, le parachute ne s'ouvre plus.

```
10 REM *****
20 REM * PARACHUTE *
30 REM *****
40 GOSUB 420
50 HH=HH-2
60 IF HH=0 THEN PUT (0,1)-(25,9)
  ,R,PSET
70 IF HH=0 THEN HH=230
80 PUT (HH,1)-(HH+25,9),H,PSET
90 D#=INKEY#
100 IF D#="" THEN 140
110 IF PV>100 THEN 140
120 IF SP=1 THEN DP=1 ELSE SP=1
130 IF DP=0 THEN PV=10:PH=HH
```



```
140 IF SP=0 THEN 230
150 IF OP=0 THEN PV=PV+4
160 IF OP=1 THEN PV=PV+1:PH=PH-1
170 IF PV>167 ORPH<1 THEN 260
180 IF OP=1 THEN 210
190 PUT (PH,PV)-(PH+14,PV+23),PF
,PSET
200 GOTO 50
210 PUT (PH,PV)-(PH+14,PV+23),PO
,PSET
220 GOTO 50
230 FOR I=1 TO 50
240 NEXT I
250 GOTO 50
260 IF ABS(PH-A)>4 THEN 320
270 FOR I=1 TO 1000
280 NEXT I
290 S=S+10
300 GOSUB 600
310 GOTO 50
320 CLS
330 FOR I=1 TO 1000
340 NEXT I
350 SCREEN 0,0
360 PRINT@ 168,"SCORE :";S
370 PRINT@ 424,"UNE AUTRE ?"
380 D#=INKEY#
390 IF D#="" THEN 380
400 IF D#<>"N" THEN RUN
410 END
420 PMODE 4
430 DIM H(25,8)
440 DIM PF(14,23)
450 DIM PO(14,23)
460 DIM R(25,8)
470 FCLS
480 DRAW "BM51,50R14L6G4L2G1L1G1
D1F1R21E1U5L2D1G2L5H1L1H1L1H1L1"
```

```
490 PAINT (58,56)
500 GET (50,50)-(75,58),H,G
510 GET (100,50)-(125,58),R,G
520 PCLS
530 DRAW "BM54,50G1L1G1D1G1D1R12
U1H1U1H1L1H1L4"
540 PAINT (56,54)
550 DRAW "BM51,56D1F1D1F1D3R2D6L
1R1U3R2D3R1L1U4L1U4L1U1R2D1L1D4R
1U2R2U3E1U1E1U1"
560 GET (50,46)-(64,69),PO,G
570 PCLS
580 DRAW "BM54,61R1D6L2D2U2R3D2L
1D1R2U1L1U2R3D2U2L3U3D3R1U6R1"
590 GET (50,46)-(64,69),PF,G
600 PCLS
610 SCREEN 1,0
620 HH=230
630 HV=1
640 A=RND(110)+10
650 DRAW "BM"+STR$(A)+"",191R12"
660 SP=0
670 DP=0
680 PV=0
690 PH=0
700 RETURN
```

LA BIBLIOTHÈQUE SYBEX

OUVRAGES GÉNÉRAUX

VOTRE PREMIER ORDINATEUR *par RODNAY ZAKS,*
296 pages, Réf. 226

VOTRE ORDINATEUR ET VOUS *par RODNAY ZAKS,*
296 pages, Réf. 271

DU COMPOSANT AU SYSTÈME : une introduction aux microprocesseurs *par RODNAY ZAKS,*
636 pages, Réf. 239

TECHNIQUES D'INTERFACE aux microprocesseurs *par AUSTIN LESEA ET RODNAY ZAKS,*
450 pages, Réf. 230, 3ème édition

LEXIQUE INTERNATIONAL MICROORDINATEURS, avec dictionnaire abrégé en 10 langues
192 pages, Réf. 234

BASIC

VOTRE PREMIER PROGRAMME BASIC *par RODNAY ZAKS,*
208 pages, Réf. 263

INTRODUCTION AU BASIC *par PIERRE LE BEUX,*
336 pages, Réf. 216

LE BASIC PAR LA PRATIQUE : 60 exercices *par JEAN-PIERRE LAMOITIER,*
252 pages, Réf. 231

LE BASIC POUR L'ENTREPRISE *par XUAN TUNG BUI,*
204 pages, Réf. 253, 2ème édition

PROGRAMMES EN BASIC, Mathématiques, Statistiques, Informatique *par ALAN R. MILLER,*
318 pages, Réf. 259

AU COEUR DES JEUX EN BASIC *par RICHARD MATEOSIAN,*
352 pages, Réf. 233

JEUX D'ORDINATEUR EN BASIC *par DAVID H. AHL,*
192 pages, Réf. 246

NOUVEAUX JEUX D'ORDINATEUR EN BASIC *par DAVID H. AHL,*
204 pages, Réf. 247

PASCAL

INTRODUCTION AU PASCAL *par PIERRE LE BEUX,*
496 pages, Réf. 222

LE PASCAL PAR LA PRATIQUE par *PIERRE LE BEUX ET HENRI TAVERNIER*,
562 pages, Réf. 229

LE GUIDE DU PASCAL par *JACQUES TIBERGHEN*,
504 pages, Réf. 232

PROGRAMMES EN PASCAL pour Scientifiques et Ingénieurs par *ALAN R. MILLER*,
392 pages, Réf. 240

AUTRES LANGAGES

INTRODUCTION A ADA par *PIERRE LE BEUX*,
366 pages, Réf. 242

MICROORDINATEUR

ALICE

JEUX EN BASIC POUR ALICE par *PIERRE MONSAUT*,
96 pages, Réf. 320

APPLE

PROGRAMMEZ EN BASIC SUR APPLE II par *LÉOPOLD LAURENT*,
208 pages, tome 1, Réf. 268

APPLE II 66 PROGRAMMES BASIC par *STANLEY R. TROST*,
192 pages, Réf. 283

JEUX EN PASCAL SUR APPLE par *DOUGLAS HERGERT ET JOSEPH T. KALASH*,
372 pages, Réf. 241

ATARI

JEUX EN BASIC SUR ATARI par *PAUL BUNN*,
96 pages, Réf. 282

COMMODORE 64

JEUX EN BASIC SUR COMMODORE 64 par *PIERRE MONSAUT*,
96 pages, Réf. 317

DRAGON

JEUX EN BASIC SUR DRAGON par *PIERRE MONSAUT*,
96 pages, Réf. 324

GOUPIL

PROGRAMMEZ VOS JEUX SUR GOUPIL par *FRANÇOIS ABELLA*,
208 pages, Réf. 264

IBM

EXERCICES EN BASIC SUR L'ORDINATEUR PERSONNEL IBM par *JEAN-PIERRE LAMOITIER*,
256 pages, Réf. 267

IBM PC Guide de l'utilisateur, *par* JOAN LASSELLE ET CAROL RAMSEY,
160 pages, Réf. 301

IBM PC 66 PROGRAMMES BASIC *par* STANLEY R. TROST,
192 pages, Réf. 280

ORIC

JEUX EN BASIC SUR ORIC *par* PETER SHAW,
96 pages, Réf. 278

SHARP

DÉCOUVREZ LE SHARP PC-1500 ET LE TRS-80 PC-2 *par* MICHEL LHOIR,
2 tomes, Réf. 261-262

SPECTRUM

PROGRAMMEZ EN BASIC SUR SPECTRUM *par* S.M. GEE,
208 pages, Réf. 252

JEUX EN BASIC SUR SPECTRUM *par* PETER SHAW,
96 pages, Réf. 276

TIMEX

DÉCOUVREZ LE ZX 81 ET LE TIMEX SINCLAIR 1000 *par* DOUGLAS HERBERT,
208 pages, Réf. 256

TRS-80

PROGRAMMEZ EN BASIC SUR TRS-80 *par* LÉOPOLD LAURENT,
2 tomes, Réf. 250-251

DÉCOUVREZ LE SHARP PC-1500 ET LE TRS-80 PC-2 *par* MICHEL LHOIR,
2 tomes, Réf. 261-262

JEUX EN BASIC SUR TRS-80 MC-10 *par* PIERRE MONSAUT,
96 pages, Réf. 323

JEUX EN BASIC SUR TRS-80 *par* CHRIS PALMER,
96 pages, Réf. 302

JEUX EN BASIC SUR TRS-80 COULEUR *par* PIERRE MONSAUT,
96 pages, Réf. 325

VIC 20

PROGRAMMEZ EN BASIC SUR VIC 20 *par* G. O. HAMANN,
2 tomes, Réf. 244-245

JEUX EN BASIC SUR VIC 20 *par* ALASTAIR GOURLAY,
96 pages, Réf. 277

ZX 81

DÉCOUVREZ LE ZX 81 ET LE TIMEX SINCLAIR 1000 *par DOUGLAS HERGERT,*
208 pages, Réf. 256

ZX 81 56 PROGRAMMES BASIC *par STANLEY R. TROST,*
192 pages, Réf. 304

GUIDE DU BASIC ZX 81 *par DOUGLAS HERGERT,*
204 pages, Réf. 285

JEUX EN BASIC SUR ZX 81 *par MARK CHARLTON,*
96 pages, Réf. 275

MICRO-PROCESSEURS

PROGRAMMATION DU Z80 *par RODNAY ZAKS,*
618 pages, Réf. 220

APPLICATIONS DU Z80 *par JAMES W. COFFRON,*
304 pages, Réf. 274

PROGRAMMATION DU 6502 *par RODNAY ZAKS,*
376 pages, Réf. 224, 2ème édition

APPLICATIONS DU 6502 *par RODNAY ZAKS,*
288 pages, Réf. 219

PROGRAMMATION DU 6800 *par DANIEL-JEAN DAVID ET RODNAY ZAKS,*
374 pages, Réf. 218

PROGRAMMATION DU 6809 *par RODNAY ZAKS ET WILLIAM LABIAK,*
392 pages, Réf. 270

SYSTÈMES D'EXPLOITATION

GUIDE DU CP/M AVEC MP/M *par RODNAY ZAKS,*
354 pages, Réf. 228

CP/M APPROFONDI *par ALAN R. MILLER,*
380 pages, Réf. 265

INTRODUCTION AU p-SYSTEM UCSD *par CHARLES W. GRANT ET JON BUTAH,*
308 pages, Réf. 257

LOGICIELS ET APPLICATIONS

INTRODUCTION AU TRAITEMENT DE TEXTE *par HAL GLATZER,*
228 pages, Réf. 243

INTRODUCTION A WORDSTAR *par ARTHUR NAIMAN,*
200 pages, Réf. 255

VISICALC APPLICATIONS *par STANLEY R. TROST,*
304 pages, Réf. 258

La plupart de ces ouvrages existent en version anglaise. N'hésitez pas à demander notre catalogue.

EN ANGLAIS

BASIC EXERCISES FOR APPLE by *JEAN-PIERRE LAMOITIER*,
232 pages, Réf. 0-084

BASIC FOR BUSINESS by *DOUGLAS HERGERT*,
224 pages, Réf. 0-080

CELESTIAL BASIC : Astronomy on your Computer by *ERIC BURGESS*,
228 pages, Réf. 0-087

INTRODUCTION TO PASCAL (Including UCSD Pascal) by *RODNEY ZAKS*,
422 pages, Réf. 0-066

DOING BUSINESS WITH PASCAL by *RICHARD HERGERT AND DOUGLAS HERGERT*,
380 pages, Réf. 0-091

MASTERING VISICALC by *DOUGLAS HERGERT*,
224 pages, Réf. 0-090

THE APPLE CONNECTION by *JAMES W. COFFRON*,
228 pages, Réf. 0-085

PROGRAMMING THE Z8000 by *RICHARD MATEOSIAN*,
300 pages, Réf. 0-032

A MICROPROGRAMMED APL IMPLEMENTATION by *RODNEY ZAKS*,
350 pages, Réf. 0-005

ADVANCED 6502 PROGRAMMING by *RODNEY ZAKS*,
292 pages, Réf. 0-089

FORTRAN PROGRAMS FOR SCIENTISTS AND ENGINEERS by *ALAN R. MILLER*,
320 pages, Réf. 0-082

***POUR UN CATALOGUE COMPLET
DE NOS PUBLICATIONS***

FRANCE
6-8, Impasse du Curé
75018 PARIS
Tél. : (1) 203.95.95
Télex : 211801

U.S.A.
2344 Sixth Street
Berkeley, CA 94710
Tel. : (415) 848.8233
Telex : 336311


ALLEMAGNE
Volgelsanger. WEG 111
4000 Düsseldorf 30
Post Bos N° 30.09.61
Tel. : (0211) 624441
Telex : 08588163



Paris • Berkeley • Düsseldorf

Achévé d'imprimer le 3 avril 1984
sur les presses de l'Imprimerie «La Source d'Or» - 63200 Marsat
Dépôt légal : 2^e trimestre 1984
Imprimeur n° 1480


1. Squash
2. Blitz
3. D.C.A.
4. Traces
5. Slalom
6. Ramasse-miettes
7. Tank
8. Alphabet
9. Atterrissage
10. Chasse au canard
11. Serpent
12. Simon
13. Grand prix
14. Crabes
15. Casse-briques
16. Poursuite
17. Parachute
18. Robots




1




2




3




4




5




6




7




8




9




10



11




12



13




14




15



16



17



18